LorentzPendulum

In[882]:=
```
Clear["Global`*"]
```

Pendulum with a given time-dependent length r[t]

In[883]:=
```
T = (1/2) * M * (r'[t]^2 + (r[t] * theta'[t])^2)
```

Out[883]= $\frac{1}{2} M \left(r'[t]^2 + r[t]^2 \, theta'[t]^2\right)$

In[884]:=
```
V = -M * g * r[t] * Cos[theta[t]]
```

Out[884]= $-g M \, Cos[theta[t]] \, r[t]$

Make small-angle approximation

In[885]:=
```
Vsmall = Normal[Series[V, {theta[t], 0, 2}]]
```

Out[885]= $-g M r[t] + \frac{1}{2} g M r[t] \, theta[t]^2$

In[886]:=
```
tmp = Expand[T - Vsmall]
```

Out[886]= $g M r[t] - \frac{1}{2} g M r[t] \, theta[t]^2 + \frac{1}{2} M r'[t]^2 + \frac{1}{2} M r[t]^2 \, theta'[t]^2$

Drop terms that contribute only constant to Integral[L*dt]

In[887]:=
```
L = tmp - Part[tmp, 1] - Part[tmp, 3]
```

Out[887]= $-\frac{1}{2} g M r[t] \, theta[t]^2 + \frac{1}{2} M r[t]^2 \, theta'[t]^2$

In[888]:=
```
ptheta = D[L, theta'[t]]
```

Out[888]= $M r[t]^2 \, theta'[t]$

In[889]:=
```
sol = Solve[ptheta == Ptheta[t], theta'[t]]
```

Out[889]= $\left\{\left\{theta'[t] \to \frac{Ptheta[t]}{M r[t]^2}\right\}\right\}$

In[890]:=
```
H = (ptheta * theta'[t] - L) /. sol[[1]]
```

Out[890]= $\frac{Ptheta[t]^2}{2 M r[t]^2} + \frac{1}{2} g M r[t] \, theta[t]^2$

Use theorem: dH/dt = partial dH/ partial dt

In[891]:= `dHdt = D[H, r[t]] * drdt`

Out[891]= $drdt\left(-\frac{Ptheta[t]^2}{M\,r[t]^3}+\frac{1}{2}g\,M\,theta[t]^2\right)$

In[892]:= `dHdr = dHdt / drdt`

Out[892]= $-\frac{Ptheta[t]^2}{M\,r[t]^3}+\frac{1}{2}g\,M\,theta[t]^2$

In[893]:= `KE = Part[H, 1]`

Out[893]= $\frac{Ptheta[t]^2}{2\,M\,r[t]^2}$

In[894]:= `PE = Part[H, 2]`

Out[894]= $\frac{1}{2}g\,M\,r[t]\,theta[t]^2$

For slowly varying r[t], KE and PE both average to HH/2 (where HH=H)

In[895]:= `sol1 = Solve[KE == HH / 2, Ptheta[t]]`

Out[895]= $\{\{Ptheta[t]\to-\sqrt{HH}\sqrt{M}\,r[t]\},\{Ptheta[t]\to\sqrt{HH}\sqrt{M}\,r[t]\}\}$

In[896]:= `sol2 = Solve[PE == HH / 2, theta[t]]`

Out[896]= $\left\{\left\{theta[t]\to-\frac{\sqrt{HH}}{\sqrt{g}\sqrt{M}\sqrt{r[t]}}\right\},\left\{theta[t]\to\frac{\sqrt{HH}}{\sqrt{g}\sqrt{M}\sqrt{r[t]}}\right\}\right\}$

In[897]:= `dHHdr = ((dHdr /. sol1[[1]]) /. sol2[[1]]) /. r[t] → r`

Out[897]= $-\frac{HH}{2\,r}$

Easy to solve d(HH)/dr = -HH/(2*r)

In[898]:= `DSolve[HH'[r] == (-1 / 2) * HH[r] / r, HH[r], r]`

Out[898]= $\left\{\left\{HH[r]\to\frac{C[1]}{\sqrt{r}}\right\}\right\}$

HH varies as 1/Sqrt[r]; so PE varies as 1/Sqrt[r], so theta^2 varies as r^(-3/2);
so amplitude in theta varies as r^(-3/4). Hence horizontal displacement x=r*theta varies as r^(1/4).

Explicit example: let r vary linearly with time.

In[899]:=
```
Hexample = H /. r[t] → v * t
```
Out[899]=
$$\frac{\text{Ptheta}[t]^2}{2\,M\,t^2\,v^2}+\frac{1}{2}\,g\,M\,t\,v\,\text{theta}[t]^2$$

In[900]:=
```
Clear[thetadot]
```

In[901]:=
```
sol = Solve[thetadot == D[Hexample, Ptheta[t]], Ptheta[t]]
```
Out[901]=
$$\{\{\text{Ptheta}[t]\to M\,t^2\,\text{thetadot}\,v^2\}\}$$

In[902]:=
```
Pth = (Ptheta[t] /. sol[[1]]) /. thetadot → theta'[t]
```
Out[902]=
$$M\,t^2\,v^2\,\text{theta}'[t]$$

In[903]:=
```
pthetadot = -D[Hexample, theta[t]]
```
Out[903]=
$$-g\,M\,t\,v\,\text{theta}[t]$$

In[904]:=

In[905]:=
```
zero = Simplify[(D[Pth, t] - pthetadot) / (M * v^2 * t)]
```
Out[905]=
$$\frac{g\,\text{theta}[t]}{v}+2\,\text{theta}'[t]+t\,\text{theta}''[t]$$

In[906]:=
```
DSolve[{(zero /. {g → 1, v → 1}) == 0, theta[10] == 1, theta'[10] == 0}, theta[t], t]
```
Out[906]=
$$\Big\{\Big\{\text{theta}[t]\to \Big(10\,\text{BesselJ}[1,2\sqrt{t}]\,\text{BesselY}[0,2\sqrt{10}]-\sqrt{10}\,\text{BesselJ}[1,2\sqrt{t}]\,\text{BesselY}[1,2\sqrt{10}]-10\,\text{BesselJ}[0,2\sqrt{10}]\,\text{BesselY}[1,2\sqrt{t}]+\sqrt{10}\,\text{BesselJ}[1,2\sqrt{10}]\,\text{BesselY}[1,2\sqrt{t}]+10\,\text{BesselJ}[2,2\sqrt{10}]\,\text{BesselY}[1,2\sqrt{t}]-10\,\text{BesselJ}[1,2\sqrt{t}]\,\text{BesselY}[2,2\sqrt{10}]\Big)\Big/\Big(\sqrt{10}\,\sqrt{t}\,\Big(\text{BesselJ}[1,2\sqrt{10}]\,\text{BesselY}[0,2\sqrt{10}]-\text{BesselJ}[0,2\sqrt{10}]\,\text{BesselY}[1,2\sqrt{10}]+\text{BesselJ}[2,2\sqrt{10}]\,\text{BesselY}[1,2\sqrt{10}]-\text{BesselJ}[1,2\sqrt{10}]\,\text{BesselY}[2,2\sqrt{10}]\Big)\Big)\Big\}\Big\}$$

In[907]:=
```
th = Simplify[theta[t] /. %[[1]]]
```
Out[907]=
$$\Big(\Big(-10\,\text{BesselJ}[0,2\sqrt{10}]+\sqrt{10}\,\text{BesselJ}[1,2\sqrt{10}]+10\,\text{BesselJ}[2,2\sqrt{10}]\Big)\,\text{BesselY}[1,2\sqrt{t}]+\text{BesselJ}[1,2\sqrt{t}]\,\Big(10\,\text{BesselY}[0,2\sqrt{10}]-\sqrt{10}\,\text{BesselY}[1,2\sqrt{10}]-10\,\text{BesselY}[2,2\sqrt{10}]\Big)\Big)\Big/\Big(\sqrt{10}\,\sqrt{t}\,\Big(\Big(-\text{BesselJ}[0,2\sqrt{10}]+\text{BesselJ}[2,2\sqrt{10}]\Big)\,\text{BesselY}[1,2\sqrt{10}]+\text{BesselJ}[1,2\sqrt{10}]\,\Big(\text{BesselY}[0,2\sqrt{10}]-\text{BesselY}[2,2\sqrt{10}]\Big)\Big)\Big)$$

In[908]:=
```
thsimple = FullSimplify[th]
```
Out[908]=
$$-10\,\pi\,\text{BesselY}[2,2\sqrt{10}]\,\text{Hypergeometric0F1}[2,-t]+\frac{1}{\sqrt{t}}50\,\pi\,\text{BesselY}[1,2\sqrt{t}]\,\text{Hypergeometric0F1}[3,-10]$$

In[909]:=
```
Plot[thsimple, {t, 10, 200}, PlotRange → All]
```

Out[909]=



In[910]:=
```
max1 = FindMaximum[thsimple, {t, 10}]
```

FindMaximum::fmgz : Encountered a gradient that is effectively zero. The result returned may not be a maximum; it may be a minimum or a saddle point. ≫

Out[910]=
```
{1., {t → 10.}}
```

In[911]:=
```
max2 = FindMaximum[thsimple, {t, 160}]
```

Out[911]=
```
{0.127069, {t → 161.23}}
```

In[912]:=
```
Solve[max2[[1]] / max1[[1]] == ((t /. max2[[2]]) / (t /. max1[[2]]))^power, power]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

Out[912]=
```
{{power → -0.742031}}
```

Close to the predicted -0.75

In[913]:=
```
Hexample = H /. r[t] → v * t
```

Out[913]=

$$\frac{\text{Ptheta}[t]^2}{2\,M\,t^2\,v^2} + \frac{1}{2}\,g\,M\,t\,v\,\text{theta}[t]^2$$

In[914]:=
```
Clear[thetadot]
```

In[915]:=
```
sol = Solve[thetadot == D[Hexample, Ptheta[t]], Ptheta[t]]
```

Out[915]=

$$\{\{\text{Ptheta}[t] \to M\,t^2\,\text{thetadot}\,v^2\}\}$$

In[916]:=
```
Pth = (Ptheta[t] /. sol[[1]]) /. thetadot → theta'[t]
```

Out[916]=

$$M\,t^2\,v^2\,\text{theta}'[t]$$

In[917]:=
```
pthetadot = -D[Hexample, theta[t]]
```

Out[917]=
```
-g M t v theta[t]
```

In[918]:=

In[919]:=
```
zero = Simplify[(D[Pth, t] - pthetadot) / (M * v^2 * t)]
```

Out[919]= $\frac{\text{g theta[t]}}{\text{v}} + 2\,\text{theta}'\text{[t]} + \text{t theta}''\text{[t]}$

In[920]:=
```
DSolve[{(zero /. {g → 1, v → 1}) == 0, theta[100] == 1, theta'[100] == 0}, theta[t], t]
```

Out[920]=
```
{{theta[t] → (10 BesselJ[1, 2 √t] BesselY[0, 20] - BesselJ[1, 2 √t] BesselY[1, 20] -
      10 BesselJ[0, 20] BesselY[1, 2 √t] + BesselJ[1, 20] BesselY[1, 2 √t] +
      10 BesselJ[2, 20] BesselY[1, 2 √t] - 10 BesselJ[1, 2 √t] BesselY[2, 20]) /
    (√t (BesselJ[1, 20] BesselY[0, 20] - BesselJ[0, 20] BesselY[1, 20] +
        BesselJ[2, 20] BesselY[1, 20] - BesselJ[1, 20] BesselY[2, 20]))}}
```

In[921]:=
```
th = Simplify[theta[t] /. %[[1]]]
```

Out[921]=
```
((-10 BesselJ[0, 20] + BesselJ[1, 20] + 10 BesselJ[2, 20]) BesselY[1, 2 √t] +
    BesselJ[1, 2 √t] (10 BesselY[0, 20] - BesselY[1, 20] - 10 BesselY[2, 20])) /
  (√t ((-BesselJ[0, 20] + BesselJ[2, 20]) BesselY[1, 20] +
      BesselJ[1, 20] (BesselY[0, 20] - BesselY[2, 20])))
```

In[922]:=
```
thsimple = FullSimplify[th]
```

Out[922]= $\frac{1}{\sqrt{t}} 100\,\pi \left(\text{BesselJ[2, 20] BesselY}\left[1, 2\sqrt{t}\right] - \text{BesselJ}\left[1, 2\sqrt{t}\right] \text{BesselY[2, 20]}\right)$

In[923]:=
```
Plot[thsimple, {t, 100, 2000}, PlotRange → All]
```

Out[923]=


In[924]:=
```
max1 = FindMaximum[thsimple, {t, 100}]
```

FindMaximum::fmgz : Encountered a gradient that is effectively zero. The result returned may not be a maximum; it may be a minimum or a saddle point. ≫

Out[924]=
```
{1., {t → 100.}}
```

In[925]:=
```
max2 = FindMaximum[thsimple, {t, 2000}]
```

Out[925]=
```
{0.106428, {t → 1988.61}}
```

In[926]:=
```
Solve[max2[[1]] / max1[[1]] == ((t /. max2[[2]]) / (t /. max1[[2]]))^power, power]
```

Solve::ifun :

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. >>

Out[926]=
```
{{power → -0.749255}}
```

Even Closer to the predicted -0.75